<?xml version="1.0" encoding="UTF-8"?>

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
	
	<xsl:include href="../admin.xsl" />
	<xsl:include href="../cp/path.xsl" />
	
	<!-- Запускаем рендер страницы -->
	<xsl:template match="/data">
		<html>
		
		<head>
			<title>
				<xsl:value-of select="$pageTitle" />
			</title>
			<xsl:call-template name="meta-head" />
		</head>
		<body>
			<xsl:call-template name="admin-header" />
			
			<div id="blocks">
				<div id="blocks_inner" />
			</div>
			
			<div id="content">
				<div id="content_inner">
					<xsl:apply-templates select="response" />
				</div>
			</div>
			
			<xsl:call-template name="footer" />
		</body>
		</html>
	</xsl:template>
	
	<xsl:template match="response">
		<div id="path">
			<xsl:apply-templates select="document('xmodule://cp/getPath/0;blogs;list_blogs')/data/response" mode="cp-path" />
		</div>
		
		<h1>Список блогов</h1>
		
		<xsl:apply-templates select="message" />
		
		<table class="standart" cellspacing="0" cellpadding="0">
			<tbody>
				<tr>
					<th>Заголовок</th>
					<th>Публикаций</th>
					<th>Читателей</th>
					<th>Рейтинг</th>
					<th>Опции</th>
				</tr>
				
				<xsl:apply-templates select="object[@type='blog']" />
			</tbody>
		</table>
		
		<script type="text/javascript">
			function delBlog(link) {
				if(confirm('Вы действительно хотите удалить этот блог?')) {
					if(confirm('Удалить публикации? Нажмите "Отмена", чтобы не удалять публикации.')) {
						link.href += '?delete_posts';
						return true;
					}
					
					return true;
				}
				
				return false;
			}
		</script>
	</xsl:template>
	
	<xsl:template match="object[@type='blog']">
		<tr>
			<td>
				<a class="mainlink" title="Перейти">
					<xsl:attribute name="href">
						<xsl:value-of select="concat($baseURI, '/blog/', property[@name='alias']/value)" />
					</xsl:attribute>
					<xsl:value-of select="property[@name='title']/value" />
				</a>
			</td>
			<td>
				<xsl:value-of select="property[@name='count_posts']/value" />
			</td>
			<td>
				<xsl:value-of select="property[@name='count_members']/value" />
			</td>
			<td>
				<xsl:value-of select="property[@name='rating']/value" />
			</td>
			<td>
				<a href="{$baseURI}/admin/blogs/delete_blog/{@id}" onclick="return delBlog(this);">удалить</a>
			</td>
		</tr>
	</xsl:template>
	
</xsl:stylesheet>